// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: contracts/LPLock.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.19;
// @custom:website App: https://alphakek.ai/
// @custom:website Telegram: https://t.me/alphakek
// @custom:website Twitter: https://twitter.com/alphakek_ai
// @notice LPLock locks Uniswap V2 LP Tokens.

import {IERC20} from "@openzeppelin/contracts/token/ERC20/IERC20.sol";

contract LPLock {
    IERC20 public constant pair =
        IERC20(0x7b97615c8efaB5D7225E0feCd1fC86f12c892e6C);
    address public constant locker = 0xDf63A5df29F1a1098Eb3D341597e811f00bdbaA9;

    bool public isTokensLocked;
    uint256 public lockedUntil;

    error OnlyLocker();
    error TokensAlreadyLocked();
    error TokensNotLocked();
    error TokensStillLocked();

    event TokensLocked(
        uint256 amount,
        uint256 startTimestamp,
        uint256 endTimestamp
    );
    event TokensUnlocked(uint256 amount, uint256 timestamp);

    modifier onlyLocker() {
        if (msg.sender != locker) {
            revert OnlyLocker();
        }
        _;
    }

    function lockTokens() external onlyLocker {
        if (isTokensLocked) {
            revert TokensAlreadyLocked();
        }

        lockedUntil = block.timestamp + 183 days;
        isTokensLocked = true;

        uint256 balance = pair.balanceOf(msg.sender);
        pair.transferFrom(msg.sender, address(this), balance);
        emit TokensLocked(balance, block.timestamp, lockedUntil);
    }

    function unlockTokens() external onlyLocker {
        if (!isTokensLocked) {
            revert TokensNotLocked();
        }

        if (block.timestamp < lockedUntil) {
            revert TokensStillLocked();
        }

        uint256 balance = pair.balanceOf(address(this));
        pair.transfer(locker, balance);

        emit TokensUnlocked(balance, block.timestamp);
    }
}